

### Example 9.4: Repeated Measures

$$Y_{ijk} = \mu + \alpha_i + S_{ij} + \tau_k + \gamma_{ik} + e_{ijk}$$

$Y_{ijk}$  strength measurement at the  
 $k$ -th time point for the  $j$ -th  
subject in the  $i$ -th program

$\alpha_i$  “fixed” program effect

$S_{ij}$  random subject effect

$\tau_k$  “fixed” time effect

$e_{ijk}$  random error

where the random effects are all  
independent and

$$\begin{aligned} S_{ij} &\sim NID(0, \sigma_S^2) \\ e_{ijk} &\sim NID(0, \sigma_\epsilon^2) \end{aligned}$$

```
/* SAS code for analyzing repeated measures data
   across time(longitudinal studies). This code
   is applied to the weightlifting data from
   Littel, et. al. (1991) This code is posted
   on the course web page under
```

```
weight2.sas      */
```

```
data set1;
  infile 'c:\stat504\weight2.dat';
  input subj program $ s1 s2 s3 s4 s5 s6 s7;
  if program='XCONT' then cprogram=3;
  if program='RI' then cprogram=1;
  if program='WI' then cprogram=2;
run;;
```

```
/* Create a data file where responses at different
   time points are on different lines */
```

```
data set2;
  set set1;
  time=2; strength=s1; output;
  time=4; strength=s2; output;
  time=6; strength=s3; output;
  time=8; strength=s4; output;
  time=10; strength=s5; output;
  time=12; strength=s6; output;
  time=14; strength=s7; output;
  keep subj program cprogram time strength;
run;
```

```

/*Create a profile plot with time on the horizontal axis*/

proc sort data=set2; by cprogram time;
run;

proc means data=set2 noprint;
  by cprogram time;
  var strength;
  output out=seta mean=strength;
run;

axis1 label=(f=swiss h=2.5)
      value=(f=swiss h=2.0) w=3.0 ;


axis2 label=(f=swiss h=2.2 a=270 r=90)
      value=(f=swiss h=2.0) w= 3.0 ;

SYMBOL1 V=circle H=2.0 w=3 l=1 i=join ;
SYMBOL2 V=diamond H=2.0 w=3 l=3 i=join ;
SYMBOL3 V=square H=2.0 w=3 l=9 i=join ;

proc gplot data=seta;
  plot strength*time=cprogram / vaxis=axis2 haxis=axis1;
  title H=3.0 F=swiss "Observed Strength Means";
  label strength=' ';
  label time = 'Time (days) ';
run;

```

## Observed Strength Means



```

/* Fit the standard mixed model with a
   compound symmetric covariance structure
   by specifying a random subject effect */

proc mixed data=set2;
  class program subj time;
  model strength = program time program*time /
    dfm=satterth;
  random subj(program) / cl;
  lsmeans program / pdiff tdiff;
  lsmeans time / pdiff tdiff;
  lsmeans program*time / slice=time pdiff tdiff;
run;

```

## The Mixed Procedure

### Model Information

Data Set	WORK.SET2
Dependent Variable	strength
Covariance Structure	Variance Components
Estimation Method	REML
Residual Variance Method	Profile
Fixed Effects SE Method	Model-Based
Degrees of Freedom Method	Satterthwaite

### Class Level Information

Class	Levels	Values
program	3	RI WI XCONT
subj	57	1 2 3 4 5 6 7 8 9 10 11 12 13 14 15 16 17 18 19 20 21 22 23 24 25 26 27 28 29 30 31 32 33 34 35 36 37 38 39 40 41 42 43 44 45 46 47 48 49 50 51 52 53 54 55 56 57
time	7	2 4 6 8 10 12 14

### Dimensions

Covariance Parameters	2
Columns in X	32
Columns in Z	57
Subjects	1
Max Obs Per Subject	399

### Number of Observations

Number of Observations Read	399
Number of Observations Used	399
Number of Observations Not Used	0

### Iteration History

Iteration	Evaluations	-2 Res Log Like	Criterion
0	1	2033.88298356	
1	1	1420.82019617	0.00000000

Convergence criteria met.

### Covariance Parameter Estimates

Cov Parm	Estimate
subj(program)	9.6033
Residual	1.1969

### Fit Statistics

-2 Res Log Likelihood	1420.8
AIC (smaller is better)	1424.8
AICC (smaller is better)	1424.9
BIC (smaller is better)	1428.9

### Solution for Random Effects

Effect			Estimate	Pred	DF	t	Pr> t
subj(prog)	RI	1	-1.4825	0.8705	81.6	-1.70	0.0923
subj(prog)	RI	2	4.2722	0.8705	81.6	4.91	<.0001
subj(prog)	RI	3	1.4650	0.8705	81.6	1.68	0.0962
.							
.							
.							
subj(prog)	XC	20	-0.2456	0.7998	90.1	-0.31	0.7595



### Type 3 Tests of Fixed Effects

Effect	Num DF	Den DF	F Value	Pr > F
program	2	54	3.07	0.0548
time	6	324	7.43	<.0001
program*time	12	324	2.99	0.0005

### Least Squares Means

Effect	Estimate	Standard Error	DF	t Value	Pr> t
program RI	80.7946	0.7816	54	103.37	<.0001
program WI	82.2245	0.6822	54	120.52	<.0001
program XCONT	79.8214	0.6991	54	114.18	<.0001
time 2	80.1617	0.4383	65.8	182.87	<.0001
time 4	80.7264	0.4383	65.8	184.16	<.0001
time 6	80.9058	0.4383	65.8	184.57	<.0001
time 8	81.1913	0.4383	65.8	185.22	<.0001
time 10	81.2230	0.4383	65.8	185.29	<.0001
time 12	81.1464	0.4383	65.8	185.12	<.0001
time 14	81.2734	0.4383	65.8	185.41	<.0001
prog*time RI 2	79.6875	0.8216	65.8	96.99	<.0001
prog*time RI 4	80.5625	0.8216	65.8	98.06	<.0001
.	.	.	.	.	.
.	.	.	.	.	.
.	.	.	.	.	.
prog*time XC 14	79.6000	0.7349	65.8	108.32	<.0001

# Differences of Least Squares Means

Level1	Level2	Estimate	Standard Error	DF	t Value	Pr> t
RI	WI	-1.4298	1.0375	54	-1.38	0.1738
RI	XCONT	0.9732	1.0486	54	0.93	0.3575
WI	XCONT	2.4031	0.9768	54	2.46	0.0171
time 2	4	-0.5647	0.2064	324	-2.74	0.0066
time 2	6	-0.7440	0.2064	324	-3.61	0.0004
.						
.						
.						
time 12	14	-0.1270	0.2064	324	-0.62	0.5388
RI 2	RI 4	-0.8750	0.3868	324	-2.26	0.0243
RI 2	RI 6	-1.1250	0.3868	324	-2.91	0.0039
.						
.						
.						
RI 4	XCONT 2	0.8125	1.1023	65.8	0.74	0.4637
.						
.						
.						
XC 12	XCONT 14	1.24E-13	0.3460	324	0.00	1.0000

### Tests of Effect Slices

Effect	time	Num DF	Den DF	F Value	Pr > F
program*time	2	2	65.8	1.08	0.3455
program*time	4	2	65.8	1.44	0.2454
program*time	6	2	65.8	1.73	0.1844
program*time	8	2	65.8	2.96	0.0590
program*time	10	2	65.8	3.77	0.0282
program*time	12	2	65.8	4.60	0.0135
program*time	14	2	65.8	5.83	0.0047

Mean strength at a particular time  
in a particular program

$$\begin{aligned}LSMEAN &= \hat{\mu} + \hat{\alpha}_i + \hat{\tau}_k + \hat{\gamma}_{ik} \\&= \bar{Y}_{i.k} = \frac{1}{n_i} \sum_{j=1}^{n_i} Y_{ijk}\end{aligned}$$

$$Var(\bar{Y}_{i.k}) = \frac{\sigma_e^2 + \sigma_S^2}{n_i}$$

$$S_{\bar{Y}_{i.k}} = \sqrt{\left(\frac{6}{7}MS_{error} + \frac{1}{7}MS_{Subj}\right)\frac{1}{n_i}}$$

↑

Cochran-Satterthwaite  
degrees of freedom are 65.8

Program means  
(averaging across time)

$$\begin{aligned}LSMEAN &= \bar{Y}_{i..} \\ &= \hat{\mu} + \hat{\alpha}_i + \frac{1}{7} \sum_{k=1}^7 (\hat{\tau}_k + \hat{\gamma}_{ik})\end{aligned}$$

$$Var(\bar{Y}_{i...}) = \frac{\sigma_e^2 + 7\sigma_S^2}{7n_i}$$

$$S_{\bar{Y}_{i..}} = \sqrt{\frac{MS_{subjects}}{2n_i}}$$

There are  $n_1 = 16$ ,  $n_2 = 21$ ,  $n_3 = 20$  subjects in the three programs.

Mean strength at a particular time point (averaging across programs)

$$\begin{aligned} LSMEAN &= \hat{\mu} + \hat{\tau}_k + \frac{1}{3} \sum_{i=1}^3 (\hat{\alpha}_i + \hat{\gamma}_{ik}) \\ &= \frac{1}{3} \sum_{i=1}^3 (\bar{Y}_{i.k}) \neq \bar{Y}_{..k} \end{aligned}$$

because  $n_1 = 16$ ,  $n_2 = 21$ ,  $n_3 = 20$ .

$$Var(LSMEAN) = \frac{1}{9} \sum_{i=1}^3 \frac{\sigma_e^2 + \sigma_S^2}{n_i}$$

$$S_{LSMEAN} = \frac{1}{3} \sqrt{\left( \frac{6}{7} MS_{error} + \frac{1}{7} MS_{subj} \right) \left( \sum_{i=1}^3 \frac{1}{n_i} \right)}$$

↑

Cochran-Satterthwaite  
degrees of freedom are 65.8

Difference between strength  
means at two time points  
(averaging across programs)

$$\begin{aligned}
 & \left( \hat{\tau}_k + \frac{1}{3} \sum_{i=1}^3 \hat{\gamma}_{ik} \right) - \left( \hat{\tau}_\ell - \frac{1}{3} \sum_{i=1}^3 \hat{\gamma}_{i\ell} \right) \\
 &= \frac{1}{3} \left( \sum_{i=1}^3 \bar{Y}_{i.k} \right) - \frac{1}{3} \left( \sum_{i=1}^3 \bar{Y}_{i.\ell} \right) \\
 &= \frac{1}{3} \sum_{i=1}^3 \frac{1}{n_i} \sum_{j=1}^{n_i} \frac{(Y_{ijk} - Y_{ij\ell})}{\phantom{Y_{ijk} - Y_{ij\ell}}}
 \end{aligned}$$

$\uparrow$   
 subject effects  
 cancel out

Variance formula:

$$\frac{2\sigma_e^2}{9} \sum_{i=1}^3 \frac{1}{n_i}$$

Standard error:

$$\sqrt{\frac{2MS_{error}}{9} \sum_{i=1}^3 \frac{1}{n_i}} = 0.206$$

↑

use degrees of freedom  
for error = 324



Difference between strength  
means for two programs  
at a specific time point

$$(\hat{\mu} + \hat{\alpha}_i + \hat{\tau}_{ik} + \hat{\gamma}_{ik}) - (\hat{\mu} + \hat{\alpha}_\ell + \hat{\tau}_k + \hat{\gamma}_{\ell k})$$

$$= \bar{Y}_{i.k} - \bar{Y}_{\ell.k}$$

$$Var(\bar{Y}_{i.k} - \bar{Y}_{\ell.k}) = (\sigma_e^2 + \sigma_S^2) \left( \frac{1}{n_i} + \frac{1}{n_\ell} \right)$$

$$S_{\bar{Y}_{i.k} - \bar{Y}_{\ell.k}} = \sqrt{\left( \frac{6}{7} MS_{error} + \frac{1}{7} MS_{subj} \right) \left( \frac{1}{n_i} + \frac{1}{n_\ell} \right)}$$

↑

Use Cochran-Satterthwaite  
degrees of freedom = 65.8

Difference between strength means at two time points within a particular program

$$(\hat{\mu} + \hat{\alpha}_i + \hat{\tau}_k + \hat{\gamma}_{ik}) - (\hat{\mu} + \hat{\alpha}_i + \hat{\tau}_\ell + \hat{\gamma}_{i\ell})$$

$$= \bar{Y}_{i.k} - \bar{Y}_{i.\ell}$$

$$Var(\bar{Y}_{i.k} - \bar{Y}_{i.\ell}) =$$

$$Var\left(\frac{1}{n_i} \sum_{j=1}^{n_i} (Y_{ijk} - Y_{ij\ell})\right) = \frac{2\sigma_e^2}{n_i}$$

$$S_{\bar{Y}_{i.k} - \bar{Y}_{i.\ell}} = \sqrt{MS_{error} \left(\frac{2}{n_i}\right)}$$

↑

use degrees of freedom  
for error=324

Difference in strength  
means for two programs  
(averaging across time points)

$$\bar{Y}_{i..} - \bar{Y}_{\ell..} = (\hat{\alpha}_i - \frac{1}{7} \sum_{k=1}^7 \hat{\gamma}_{ik}) - (\hat{\alpha}_{\ell} + \frac{1}{7} \sum_{k=1}^7 \hat{\gamma}_{\ell k})$$

$$Var(\bar{Y}_{i..} - \bar{Y}_{\ell..}) = (\sigma_e^2 + 7\sigma_S^2)(\frac{1}{7n_i} + \frac{1}{7n_{\ell}})$$

$$S_{\bar{Y}_{i..} - \bar{Y}_{\ell..}} = \sqrt{\underset{\substack{\uparrow \\ 54 \text{ d.f.}}}{MS_{subjects}} (\frac{1}{7n_i} + \frac{1}{7n_{\ell}})}$$

```
/* Use the GLM procedure in SAS to get formulas for
   expectations of mean squares.  */

proc glm data=set2;
  class program subj time;
  model strength = program subj(program) time program*time;
  random subj(program);
  lsmeans program / pdiff tdiff;
  lsmeans time / pdiff tdiff;
  lsmeans program*time / slice=time pdiff tdiff;
run;
```

## The GLM Procedure

### Class Level Information

Class	Levels	Values
program	3	RI WI XCONT
subj	57	1 2 3 4 5 6 7 8 9 10 11 12 13 14 15 16 17 18 19 20 21 22 23 24 25 26 27 28 29 30 31 32 33 34 35 36 37 38 39 40 41 42 43 44 45 46 47 48 49 50 51 52 53 54 55 56 57
time	7	2 4 6 8 10 12 14

Number of Observations Read	399
Number of Observations Used	399

Dependent Variable: strength

Source	DF	Sum of Squares	Mean Square	F	Pr > F
Model	74	4210.0529	56.8926	47.53	<.0001
Error	324	387.7867	1.1969		
C Total	398	4597.8396			

Source	DF	Type III SS	Mean Square	F	Pr > F
program	2	419.4352	209.7176	175.22	<.0001
subj(program)	54	3694.6900	68.4202	57.17	<.0001
time	6	52.9273	8.8212	7.37	<.0001
program*time	12	43.0002	3.5834	2.99	0.0005

Source	Type III Expected Mean Square
program	Var(Error) + 7 Var(subj(program)) + Q(program,program*time)
subj(program)	Var(Error) + 7 Var(subj(program))
time	Var(Error) + Q(time,program*time)
program*time	Var(Error) + Q(program*time)

## Specifying other covariance matrices

We began with the model

$$Y_{ijk} = \mu + \alpha_i + S_{ij} + \tau_k + \gamma_{ik} + e_{ijk}$$

where

$$S_{ij} \sim NID(0, \sigma_S^2)$$

$$e_{ijk} \sim NID(0, \sigma_e^2)$$

and the  $\{S_{ij}\}$  are distributed independently of the  $\{e_{ijk}\}$

This model was expressed in the form

$$\mathbf{Y} = \mathbf{X}\boldsymbol{\beta} + \mathbf{Z}\mathbf{u} + \mathbf{e}$$

where

$$\mathbf{u} = \begin{bmatrix} S_{11} \\ \vdots \\ S_{3,20} \end{bmatrix}$$

contained the random subject effects



Here

$$G = Var(\mathbf{u}) = \sigma_S^2 I$$

$$R = Var(\mathbf{e}) = \sigma_e^2 I$$

$$\Sigma = Var(\mathbf{Y}) = ZGZ^T + R$$

$$= \sigma_S^2 \begin{bmatrix} J & & & \\ & J & & \\ & & \dots & \\ & & & J \end{bmatrix} + \sigma_e^2 I$$

$$= \begin{bmatrix} \sigma_e^2 I + \sigma_S^2 J & & & \\ & \dots & & \\ & & \dots & \\ & & & \sigma_e^2 I + \sigma_S^2 J \end{bmatrix}$$

where  $J$  is a matrix of ones.

If you are not interested in predicting subject effects (random subject effects are included only to introduce correlation among repeated measures on the same subject), you can work with an alternative expression of the same model

$$\mathbf{Y} = X\boldsymbol{\beta} + \mathbf{e}^*$$

where

$$\begin{aligned} R &= \text{Var}(\mathbf{e}^*) \\ &= \begin{bmatrix} \sigma_e^2 I + \sigma_S^2 J & & \\ & \dots & \\ & & \sigma_e^2 I + \sigma_S^2 J \end{bmatrix} \end{aligned}$$

Replace the mixed model

$$\mathbf{Y} = X\boldsymbol{\beta} + Z\mathbf{u} + \mathbf{e}$$

with the model

$$\mathbf{Y} = X\boldsymbol{\beta} + \mathbf{e}^*$$

where

$$Var(\mathbf{Y}) = Var(\mathbf{e}^*) = \begin{bmatrix} W & 0 & \cdots & 0 \\ 0 & W & \cdots & 0 \\ \cdots & \cdots & \cdots & \cdots \\ 0 & 0 & \cdots & W \end{bmatrix}$$

You can specify this model by using  
the REPEATED statement in  
PROC MIXED

REPEATED / type =

subject = subj(program)

↗ ↗  
variables in the  
class statement

$r$   
↗  
print the  
 $W$   
matrix for  
one subject

rcorr;  
↗  
print the  
correlation  
matrix for  
one subject

**Compound Symmetry:** (type = CS)

$$W = \begin{bmatrix} \sigma_1^2 + \sigma_2^2 & \sigma_2^2 & \sigma_2^2 & \sigma_2^2 \\ \sigma_2^2 & \sigma_1^2 + \sigma_2^2 & \sigma_2^2 & \sigma_2^2 \\ \sigma_2^2 & \sigma_2^2 & \sigma_1^2 + \sigma_2^2 & \sigma_2^2 \\ \sigma_2^2 & \sigma_2^2 & \sigma_2^2 & \sigma_1^2 + \sigma_2^2 \end{bmatrix}$$

**Variance components:** (type = VC)  
(default)

$$W = \begin{bmatrix} \sigma_1^2 & 0 & 0 & 0 \\ 0 & \sigma_2^2 & 0 & 0 \\ 0 & 0 & \sigma_3^2 & 0 \\ 0 & 0 & 0 & \sigma_4^2 \end{bmatrix}$$

**Unstructured:** (type = UN)

$$W = \begin{bmatrix} \sigma_1^2 & \sigma_{12} & \sigma_{13} & \sigma_{14} \\ \sigma_{12} & \sigma_2^2 & \sigma_{23} & \sigma_{24} \\ \sigma_{13} & \sigma_{23} & \sigma_3^2 & \sigma_{34} \\ \sigma_{14} & \sigma_{24} & \sigma_{34} & \sigma_4^2 \end{bmatrix}$$

**Toeplitz:** (type = TOEP)

$$W = \begin{bmatrix} \sigma^2 & \sigma_1 & \sigma_2 & \sigma_3 \\ \sigma_1 & \sigma^2 & \sigma_1 & \sigma_2 \\ \sigma_2 & \sigma_1 & \sigma^2 & \sigma_1 \\ \sigma_3 & \sigma_2 & \sigma_1 & \sigma^2 \end{bmatrix}$$

## Heterogeneous Toeplitz:

(type = TOEPH)

$$W = \begin{bmatrix} \sigma_1^2 & \sigma_1\sigma_2\rho_1 & \sigma_1\sigma_2\rho_2 & \sigma_1\sigma_4\rho_3 \\ \sigma_2\sigma_1\rho_1 & \sigma_2^2 & \sigma_2\sigma_3\rho_1 & \sigma_2\sigma_4\rho_2 \\ \sigma_3\sigma_1\rho_2 & \sigma_3\sigma_2\rho_1 & \sigma_3^2 & \sigma_3\sigma_4\rho_1 \\ \sigma_4\sigma_1\rho_3 & \sigma_4\sigma_2\rho_2 & \sigma_4\sigma_3\rho_1 & \sigma_4^2 \end{bmatrix}$$

## First order Ante-dependence:

(type = ANTE(1))

$$W = \begin{bmatrix} \sigma_1^2 & \sigma_1\sigma_2\rho_1 & \sigma_1\sigma_3\rho_1\rho_2 \\ \sigma_2\sigma_1\rho_1 & \sigma_2^2 & \sigma_2\sigma_3\rho_2 \\ \sigma_3\sigma_1\rho_2\rho_1 & \sigma_3\sigma_2\rho_2 & \sigma_3^2 \end{bmatrix}$$

**First Order Autoregressive:**  
(type = AR(1))


$$W = \sigma^2 \begin{bmatrix} 1 & \rho & \rho^2 & \rho^3 \\ \rho & 1 & \rho & \rho^2 \\ \rho^2 & \rho & 1 & \rho \\ \rho^3 & \rho^2 & \rho & 1 \end{bmatrix}$$

**Heterogeneous AR(1):** (type = ARH(1))

$$W = \begin{bmatrix} \sigma_1^2 & \sigma_1\sigma_2\rho & \sigma_1\sigma_3\rho^2 & \sigma_1\sigma_4\rho^3 \\ \sigma_2\sigma_1\rho & \sigma_2^2 & \sigma_2\sigma_3\rho & \sigma_2\sigma_4\rho^2 \\ \sigma_3\sigma_1\rho^2 & \sigma_3\sigma_2\rho & \sigma_3^2 & \sigma_3\sigma_4\rho \\ \sigma_4\sigma_1\rho^3 & \sigma_4\sigma_2\rho^2 & \sigma_4\sigma_3\rho & \sigma_4^2 \end{bmatrix}$$



**Spatial power:** (type = sp(pow)(list))

list of variables  
defining coordinates

$$W = \sigma^2 \begin{bmatrix} 1 & \rho^{d_{12}} & \rho^{d_{13}} & \rho^{d_{14}} \\ \rho^{d_{12}} & 1 & \rho^{d_{23}} & \rho^{d_{24}} \\ \rho^{d_{13}} & \rho^{d_{23}} & 1 & \rho^{d_{34}} \\ \rho^{d_{14}} & \rho^{d_{24}} & \rho^{d_{34}} & 1 \end{bmatrix}$$

where  $d_{ij}$  is the Euclidean distance  
between the  $i$ -th and  $j$ -th observations  
provided by one subject (or unit).

You can replace *pow* with a number of other  
choices.

## Selecting a Covariance Structure

Assume  $E(\mathbf{Y}) = X\boldsymbol{\beta}$  is correct.

- Likelihood ratio tests for “nested” models

$$\begin{aligned} & -2 \left( \begin{array}{l} \text{REML log-likelihood for} \\ \text{the smaller model} \end{array} \right) \\ & - \left[ -2 \left( \begin{array}{l} \text{REML log-likelihood for} \\ \text{the larger model} \end{array} \right) \right] \\ & \quad \sim \chi_{df}^2 \quad \text{for large } n \end{aligned}$$

where

$$\begin{aligned} df = & \left[ \begin{array}{l} \text{number of covariance} \\ \text{parameters in the} \\ \text{larger model} \end{array} \right] \\ & - \left[ \begin{array}{l} \text{number of covariance} \\ \text{parameters in the} \\ \text{smaller model} \end{array} \right] \end{aligned}$$

Consider models with

- Larger values of the Akaike Information Criterion (AIC)

(REML Log-likelihood)

$$- \left( \begin{array}{l} \text{number of parameters in} \\ \text{the covariance model} \end{array} \right)$$

- Larger values of the Schwarz Bayesian Criterion (SBC)

(REML log-likelihood)

$$- \frac{\log(n-p)}{2} \left( \begin{array}{l} \text{number of parameters} \\ \text{in the covariance model} \end{array} \right)$$

Here  $n = 7 \times 57 = 399$  observations

$p = 21$  parameters in  $X\beta$

```
/* Fit the same model with the repeated statement.  
   Here the compound symmetry covariance structure  
   is selected. */
```

```
proc mixed data=set2;  
  class program subj time;  
  model strength = program time program*time;  
  repeated / type=cs sub=subj(program) r rcorr;  
run;
```

```
/* Fit a model with the same fixed effects,  
   but change the covariance structure to  
   an AR(1) model */
```

```
proc mixed data=set2;  
  class program subj time;  
  model strength = program time program*time;  
  repeated / type=ar(1) sub=subj(program) r rcorr;  
run;
```

## The Mixed Procedure

### Model Information

Data Set	WORK.SET2
Dependent Variable	strength
Covariance Structure	Compound Symmetry
Subject Effect	subj(program)
Estimation Method	REML
Residual Variance Method	Profile
Fixed Effects SE Method	Model-Based
Degrees of Freedom Method	Between-Within

### Class Level Information

Class	Levels	Values
program	3	RI WI XCONT
subj	57	1 2 3 4 5 6 7 8 9 10 11 12 13 14 15 16 17 18 19 20 21 22 23 24 25 26 27 28 29 30 31 32 33 34 35 36 37 38 39 40 41 42 43 44 45 46 47 48 49 50 51 52 53 54 55 56 57
time	7	2 4 6 8 10 12 14

### Iteration History

Iteration	Evaluations	-2 Res Log Like	Criterion
0	1	2033.88298356	
1	1	1420.82019617	0.00000000

Estimated R Matrix for subj(program) 1 RI

Row	Col1	Col2	Col3	Col4	Col5	Col6	Col7
1	10.8002	9.6033	9.6033	9.6033	9.6033	9.6033	9.6033
2	9.6033	10.8002	9.6033	9.6033	9.6033	9.6033	9.6033
3	9.6033	9.6033	10.8002	9.6033	9.6033	9.6033	9.6033
4	9.6033	9.6033	9.6033	10.8002	9.6033	9.6033	9.6033
5	9.6033	9.6033	9.6033	9.6033	10.8002	9.6033	9.6033
6	9.6033	9.6033	9.6033	9.6033	9.6033	10.8002	9.6033
7	9.6033	9.6033	9.6033	9.6033	9.6033	9.6033	10.8002

Estimated R Correlation Matrix for subj(program) 1 RI

Row	Col1	Col2	Col3	Col4	Col5	Col6	Col7
1	1.0000	0.8892	0.8892	0.8892	0.8892	0.8892	0.8892
2	0.8892	1.0000	0.8892	0.8892	0.8892	0.8892	0.8892
3	0.8892	0.8892	1.0000	0.8892	0.8892	0.8892	0.8892
4	0.8892	0.8892	0.8892	1.0000	0.8892	0.8892	0.8892
5	0.8892	0.8892	0.8892	0.8892	1.0000	0.8892	0.8892
6	0.8892	0.8892	0.8892	0.8892	0.8892	1.0000	0.8892
7	0.8892	0.8892	0.8892	0.8892	0.8892	0.8892	1.0000

### Covariance Parameter Estimates

Cov Parm	Subject	Estimate
CS	subj(program)	9.6033
Residual		1.1969

### Fit Statistics

-2 Res Log Likelihood	1420.8
AIC (smaller is better)	1424.8
AICC (smaller is better)	1424.9
BIC (smaller is better)	1428.9

### Type 3 Tests of Fixed Effects

Effect	Num DF	Den DF	F Value	Pr > F
program	2	54	3.07	0.0548
time	6	324	7.43	<.0001
program*time	12	324	2.99	0.0005

## The Mixed Procedure

### Model Information

Data Set	WORK.SET2
Dependent Variable	strength
Covariance Structure	Autoregressive
Subject Effect	subj(program)
Estimation Method	REML
Residual Variance Method	Profile
Fixed Effects SE Method	Model-Based
Degrees of Freedom Method	Between-Within

### Class Level Information

Class	Levels	Values
program	3	RI WI XCONT
subj	57	1 2 3 4 5 6 7 8 9 10 11 12 13 14 15 16 17 18 19 20 21 22 23 24 25 26 27 28 29 30 31 32 33 34 35 36 37 38 39 40 41 42 43 44 45 46 47 48 49 50 51 52 53 54 55 56 57
time	7	2 4 6 8 10 12 14

### Iteration History

Iteration	Evaluations	-2 Res Log Like	Criterion
0	1	2033.88298356	
1	2	1266.80350600	0.00000002
2	1	1266.80350079	0.00000000



Estimated R Matrix for subj(program) 1 RI

Row	Col1	Col2	Col3	Col4	Col5	Col6	Col7
1	10.7600	10.2411	9.7473	9.2772	8.8298	8.4040	7.9988
2	10.2411	10.7600	10.2411	9.7473	9.2772	8.8298	8.4040
3	9.7473	10.2411	10.7600	10.2411	9.7473	9.2772	8.8298
4	9.2772	9.7473	10.2411	10.7600	10.2411	9.7473	9.2772
5	8.8298	9.2772	9.7473	10.2411	10.7600	10.2411	9.7473
6	8.4040	8.8298	9.2772	9.7473	10.2411	10.7600	10.2411
7	7.9988	8.4040	8.8298	9.2772	9.7473	10.2411	0.7600

Estimated R Correlation Matrix for subj(program) 1 RI

Row	Col1	Col2	Col3	Col4	Col5	Col6	Col7
1	1.0000	0.9518	0.9059	0.8622	0.8206	0.7810	0.7434
2	0.9518	1.0000	0.9518	0.9059	0.8622	0.8206	0.7810
3	0.9059	0.9518	1.0000	0.9518	0.9059	0.8622	0.8206
4	0.8622	0.9059	0.9518	1.0000	0.9518	0.9059	0.8622
5	0.8206	0.8622	0.9059	0.9518	1.0000	0.9518	0.9059
6	0.7810	0.8206	0.8622	0.9059	0.9518	1.0000	0.9518
7	0.7434	0.7810	0.8206	0.8622	0.9059	0.9518	1.0000

### Covariance Parameter Estimates

Cov Parm	Subject	Estimate
AR(1)	subj(program)	0.9518
Residual		10.7600

### Fit Statistics

-2 Res Log Likelihood	1266.8
AIC (smaller is better)	1270.8
AICC (smaller is better)	1270.8
BIC (smaller is better)	1274.9

### Type 3 Tests of Fixed Effects

Effect	Num DF	Den DF	F Value	Pr > F
program	2	54	3.11	0.0528
time	6	324	4.30	0.0003
program*time	12	324	1.17	0.3007

```
/* Fit a model with the same fixed effects,  
   but use an arbitrary covariance matrix  
   for repated measures on the same subject. */  
  
proc mixed data=set2;  
  class program subj time;  
  model strength = program time program*time;  
  repeated / type=un sub=subj(program) r rcorr;  
run;
```

## The Mixed Procedure

### Model Information

Data Set	WORK.SET2
Dependent Variable	strength
Covariance Structure	Unstructured
Subject Effect	subj(program)
Estimation Method	REML
Residual Variance Method	None
Fixed Effects SE Method	Model-Based
Degrees of Freedom Method	Between-Within

### Class Level Information

Class	Levels	Values
program	3	RI WI XCONT
subj	57	1 2 3 4 5 6 7 8 9 10 11 12 13 14 15 16 17 18 19 20 21 22 23 24 25 26 27 28 29 30 31 32 33 34 35 36 37 38 39 40 41 42 43 44 45 46 47 48 49 50 51 52 53 54 55 56 57
time	7	2 4 6 8 10 12 14

### Iteration History

Iteration	Evaluations	-2 Res Log Like	Criterion
0	1	2033.882984	
1	1	1234.895726	0.00000000

Estimated R Matrix for subj(program) 1 RI

Row	Col1	Col2	Col3	Col4	Col5	Col6	Col7
1	8.7804	8.7573	8.9659	8.1986	8.6784	8.2206	8.4172
2	8.7573	9.4732	9.4633	8.5688	9.2015	8.7310	8.6878
3	8.9659	9.4633	10.7083	9.9268	10.6664	10.0704	10.2142
4	8.1986	8.5688	9.9268	10.0776	10.5998	9.8989	10.0436
5	8.6784	9.2015	10.6664	10.5998	12.0954	11.3447	11.3641
6	8.2206	8.7310	10.0704	9.8989	11.3447	11.7562	11.6504
7	8.4172	8.6878	10.2142	10.0436	11.3641	11.6504	12.7104

Estimated R Correlation Matrix for subj(program) 1 RI

Row	Col1	Col2	Col3	Col4	Col5	Col6	Col7
1	1.0000	0.9602	0.9246	0.8716	0.8421	0.8091	0.7968
2	0.9602	1.0000	0.9396	0.8770	0.8596	0.8273	0.7917
3	0.9246	0.9396	1.0000	0.9556	0.9372	0.8975	0.8755
4	0.8716	0.8770	0.9556	1.0000	0.9601	0.9094	0.8874
5	0.8421	0.8596	0.9372	0.9601	1.0000	0.9514	0.9165
6	0.8091	0.8273	0.8975	0.9094	0.9514	1.0000	0.9531
7	0.7968	0.7917	0.8755	0.8874	0.9165	0.9531	1.0000

### Fit Statistics

-2 Res Log Likelihood	1234.9
AIC (smaller is better)	1290.9
AICC (smaller is better)	1295.5
BIC (smaller is better)	1348.1

### Type 3 Tests of Fixed Effects

Effect	Num DF	Den DF	F Value	Pr > F
program	2	54	3.07	0.0548
time	6	54	7.12	<.0001
program*time	12	54	1.57	0.1297

Covariance Model	-2 REML log-like.	AIC
Compound Symmetry (2 parms)	1420.8	1424.8
AR(1) (2 parms)	1266.8	1270.8
Unstructured (28 parms)	1234.9	1290.9

- The AR(1) covariance structure is indicated
- Results may change if the “fixed” part of the model ( $X\beta$ ) is changed

### Likelihood ratio tests:

1.  $H_0$ : compound symmetry vs.  
 $H_A$ : unstructured

$$(1420.820) - (1234.896) = 185.924$$

on  $(28)-(2) = 26$  d.f.

(p-value = .0001)

2.  $H_0$ : AR(1) vs.  $H_A$ : unstructured

$$(1266.804) - (1234.896) = 31.908$$

on  $(28)-(2)=26$  d.f.

(p-value = 0.196)

3. Do not use this model to test

$H_0$ : AR(1) vs.  $H_A$ : compound symmetry



```

/* Fit a model with linear and quadratic
   trends across time and different trends
   across time for different programs. */

proc mixed data=set2;
  class program subj;
  model strength = program time time*program
    time*time time*time*program / htype=1;
  repeated / type=ar(1) sub=subj(program);
run;

/* Fit a model with linear, quadratic and
   cubic trends across time and different
   trends across time for different programs. */

proc mixed data=set2;
  class program subj;
  model strength = program time time*program
    time*time time*time*program
    time*time*time time*time*time*program / htype=1;
  repeated / type=ar(1) sub=subj(program);
run;

```

```

/* By removing the automatic intercept and adding
   the solution option to the model statement, the
   estimates of the parameters in the model are obtained.
   Here we use a power function model for the covariance
   structure. This is a generalization of the AR(1)
   covariance structure that can be used with unequally
   spaced time points. */

proc mixed data=set2;
  class program subj;
  model strength = program time*program time*time*program
                    / noint solution htype=1;
  repeated / type=sp(pow)(time) sub=subj(program) r rcorr;
run;

```

## The Mixed Procedure

### Model Information

Data Set	WORK.SET2
Dependent Variable	strength
Covariance Structure	Autoregressive
Subject Effect	subj(program)
Estimation Method	REML
Residual Variance Method	Profile
Fixed Effects SE Method	Model-Based
Degrees of Freedom Method	Between-Within

### Class Level Information

Class	Levels	Values
program	3	RI WI XCONT
subj	57	1 2 3 4 5 6 7 8 9 10 11 12 13 14 15 16 17 18 19 20 21 22 23 24 25 26 27 28 29 30 31 32 33 34 35 36 37 38 39 40 41 42 43 44 45 46 47 48 49 50 51 52 53 54 55 56 57

### Iteration History

Iteration	Evaluations	-2 Res Log Like	Criterion
0	1	2090.993400	
1	2	1293.507056	0.00000122
2	1	1293.506703	0.00000000

### Covariance Parameter Estimates

Cov Parm	Subject	Estimate
AR(1)	subj(program)	0.9523
Residual		10.7585

### Fit Statistics

-2 Res Log Likelihood	1293.5
AIC (smaller is better)	1297.5
AICC (smaller is better)	1297.5
BIC (smaller is better)	1301.6

### Type 1 Tests of Fixed Effects

Effect	Num DF	Den DF	F Value	Pr > F
program	2	54	3.10	0.0530
time	1	336	12.69	0.0004
time*program	2	336	4.75	0.0093
time*time	1	336	7.18	0.0077
time*time*program	2	336	0.88	0.4167

## The Mixed Procedure

### Model Information

Data Set	WORK.SET2
Dependent Variable	strength
Covariance Structure	Autoregressive
Subject Effect	subj(program)
Estimation Method	REML
Residual Variance Method	Profile
Fixed Effects SE Method	Model-Based
Degrees of Freedom Method	Between-Within

### Class Level Information

Class	Levels	Values
program	3	RI WI XCONT
subj	57	1 2 3 4 5 6 7 8 9 10 11 12 13 14 15 16 17 18 19 20 21 22 23 24 25 26 27 28 29 30 31 32 33 34 35 36 37 38 39 40 41 42 43 44 45 46 47 48 49 50 51 52 53 54 55 56 57

### Iteration History

Iteration	Evaluations	-2 Res Log Like	Criterion
0	1	2115.710057	
1	2	1323.826724	0.00004384
2	1	1323.813124	0.00000002
3	1	1323.813118	0.00000000

### Covariance Parameter Estimates

Cov Parm	Subject	Estimate
AR(1)	subj(program)	0.9522
Residual		10.7590

### Fit Statistics

-2 Res Log Likelihood	1323.8
AIC (smaller is better)	1327.8
AICC (smaller is better)	1327.8
BIC (smaller is better)	1331.9

### Type 1 Tests of Fixed Effects

Effect	Num DF	Den DF	F Value	Pr > F
program	2	54	3.10	0.0530
time	1	333	8.85	0.0031
time*program	2	333	4.39	0.0131
time*time	1	333	7.18	0.0077
time*time*program	2	333	0.88	0.4170
time*time*time	1	333	2.72	0.1001
time*time*time*program	2	333	0.03	0.9740

## The Mixed Procedure

### Model Information

Data Set	WORK.SET2
Dependent Variable	strength
Covariance Structure	Spatial Power
Subject Effect	subj(program)
Estimation Method	REML
Residual Variance Method	Profile
Fixed Effects SE Method	Model-Based
Degrees of Freedom Method	Between-Within

### Class Level Information

Class	Levels	Values
program	3	RI WI XCONT
subj	57	1 2 3 4 5 6 7 8 9 10 11 12 13 14 15 16 17 18 19 20 21 22 23 24 25 26 27 28 29 30 31 32 33 34 35 36 37 38 39 40 41 42 43 44 45 46 47 48 49 50 51 52 53 54 55 56 57

### Iteration History

Iteration	Evaluations	-2 Res Log Like	Criterion
0	1	2090.993400	
1	2	1314.086838	0.11555121
2	1	1293.541779	0.00012430
3	1	1293.506742	0.00000014
4	1	1293.506703	0.00000000

Estimated R Matrix for subj(program) 1 RI

Row	Col1	Col2	Col3	Col4	Col5	Col6	Col7
1	10.7584	10.2449	9.7559	9.2902	8.8468	8.4245	8.0224
2	10.2449	10.7584	10.2449	9.7559	9.2902	8.8468	8.4245
3	9.7559	10.2449	10.7584	0.2449	9.7559	9.2902	8.8468
4	9.2902	9.7559	10.2449	10.7584	10.2449	9.7559	9.2902
5	8.8468	9.2902	9.7559	10.2449	10.7584	10.2449	9.7559
6	8.4245	8.8468	9.2902	9.7559	10.2449	10.7584	10.2449
7	8.0224	8.4245	8.8468	9.2902	9.7559	10.2449	10.7584

Estimated R Correlation Matrix for subj(program) 1 RI

Row	Col1	Col2	Col3	Col4	Col5	Col6	Col7
1	1.0000	0.9523	0.9068	0.8635	0.8223	0.7831	0.7457
2	0.9523	1.0000	0.9523	0.9068	0.8635	0.8223	0.7831
3	0.9068	0.9523	1.0000	0.9523	0.9068	0.8635	0.8223
4	0.8635	0.9068	0.9523	1.0000	0.9523	0.9068	0.8635
5	0.8223	0.8635	0.9068	0.9523	1.0000	0.9523	0.9068
6	0.7831	0.8223	0.8635	0.9068	0.9523	1.0000	0.9523
7	0.7457	0.7831	0.8223	0.8635	0.9068	0.9523	1.0000



### Covariance Parameter Estimates

Cov Parm	Subject	Estimate
SP(POW)	subj(program)	0.9758
Residual		10.7584

### Fit Statistics

-2 Res Log Likelihood	1293.5
AIC (smaller is better)	1297.5
AICC (smaller is better)	1297.5
BIC (smaller is better)	1301.6

### Solution for Fixed Effects

Effect	program	Estimate	Standard Error	Pr> t
program	RI	78.9054	0.8913	<.0001
program	WI	80.4928	0.7780	<.0001
program	XCONT	79.5708	0.7972	<.0001
time*program	RI	0.4303	0.1315	0.0012
time*program	WI	0.2930	0.1148	0.0111
time*program	XCONT	0.1046	0.1176	0.3746
time*time*program	RI	-0.01942	0.007634	0.0114
time*time*program	WI	-0.00766	0.006664	0.2514
time*time*program	XCONT	-0.00732	0.006828	0.2842

### Type 1 Tests of Fixed Effects

Effect	Num DF	Den DF	F Value	Pr > F
program	3	54	12910.9	<.0001
time*program	3	336	7.39	<.0001
time*time*program	3	336	2.98	0.0316

## Models for the change in mean strength across time

### Program 1: Repetitions increase

$$Y = 78.9054 + 0.4303(\text{days}) - .01942(\text{days})^2$$

(0.8913)    (0.1315)            (0.0076)

### Program 2: Weight increases

$$Y = 80.4928 + 0.2930(\text{days}) - .00765(\text{days})^2$$

(0.7779)    (0.1148)            (.00666)

### Program 3: Controls

$$Y = 79.5708 + 0.1046(\text{days}) - .00732(\text{days})^2$$

(0.7972)    (0.1176)            (.00683)

```

/* Fit a model with random coefficients that allow
   individual subjects to have different linear and
   quadratic trends across time. */

proc mixed data=set2 scoring=8;
  class program subj;
  model strength = program time*program time*time*program/
                    noint solution htype=1;
  random int time time*time / type=un sub=subj(program)
                    solution ;
run;

```

## The Mixed Procedure

### Model Information

Data Set	WORK.SET2
Dependent Variable	strength
Covariance Structure	Unstructured
Subject Effect	subj(program)
Estimation Method	REML
Residual Variance Method	Profile
Fixed Effects SE Method	Model-Based
Degrees of Freedom Method	Containment

### Class Level Information

Class	Levels	Values
program	3	RI WI XCONT
subj	57	1 2 3 4 5 6 7 8 9 10 11 12 13 14 15 16 17 18 19 20 21 22 23 24 25 26 27 28 29 30 31 32 33 34 35 36 37 38 39 40 41 42 43 44 45 46 47 48 49 50 51 52 53 54 55 56 57

### Iteration History

Iteration	Evaluations	-2 Res Log Like	Criterion
0	1	2090.99340	
1	1	1304.01337	0.0000

Convergence criteria met.

### Covariance Parameter Estimates

Cov Parm	Subject	Estimate
UN(1,1)	subj(program)	9.3596
UN(2,1)	subj(program)	-0.3437
UN(2,2)	subj(program)	0.1868
UN(3,1)	subj(program)	0.01417
UN(3,2)	subj(program)	-0.00942
UN(3,3)	subj(program)	0.000575
Residual		0.4518

### Fit Statistics

-2 Res Log Likelihood	1304.0
AIC (smaller is better)	1318.0
AICC (smaller is better)	1318.3
BIC (smaller is better)	1332.3

### Solution for Fixed Effects

Effect	program	Estimate	Standard Error	Pr> t
program	RI	79.0804	0.8084	<.0001
program	WI	80.5238	0.7056	<.0001
program	XCONT	79.6071	0.7231	<.0001
time*program	RI	0.3966	0.1315	0.0039
time*program	WI	0.3005	0.1148	0.0115
time*program	XCONT	0.1116	0.1177	0.3471
time*time*program	RI	-0.01823	0.007547	0.0191
time*time*program	WI	-0.00879	0.006587	0.1878
time*time*program	XCONT	-0.00848	0.006750	0.2143

### Type 1 Tests of Fixed Effects

Effect	Num DF	Den DF	F Value	Pr > F
program	3	54	12749.5	<.0001
time*program	3	54	6.57	0.0007
time*time*program	3	54	3.06	0.0356



# Solution for Random Effects

Effect	prog	subj	Estimate	Std Err Pred	DF	Pr> t
Intercept	RI	1	-0.5827	1.1418	228	0.6103
time	RI	1	-0.1998	0.2551	228	0.4343
time*time	RI	1	0.008501	0.01522	228	0.5771
Intercept	RI	2	2.5097	1.1418	228	0.0290
time	RI	2	0.1939	0.2551	228	0.4481
time*time	RI	2	0.003286	0.01522	228	0.8293
Intercept	RI	3	1.4849	1.1418	228	0.1947
time	RI	3	0.04297	0.2551	228	0.8664
time*time	RI	3	-0.00436	0.01522	228	0.7749
.						
.						
.						
Intercept	XC	19	0.4741	1.0937	228	0.6651
time	XC	19	-0.2557	0.2519	228	0.3112
time*time	XC	19	0.01829	0.01507	228	0.2262
Intercept	XC	20	-2.1431	1.0937	228	0.0513
time	XC	20	0.4422	0.2519	228	0.0806
time*time	XC	20	-0.02043	0.01507	228	0.1765